
Using of FORTRAN 77 FILES

A FILE is necessary when
- the data must be retained between 2 executions ;
- the volume of data is so great than the size of the main memory is not sufficient !

In this last case, the large data is subdivided into records or pages of which only a
small number of its are stored in main memory and all others are stored on secondary
memories as disks, USB cards, ... This set of pages constitutes one file.

As one goes along needs, the pages succeed in MM (Main Memory), unique place where
they may be initialized or modified. The programmer decides to READ a page (transfer
from the disk SM (Secondary Memory) to the MM under the form of variables, arrays,
... , modify their contains and WRITE them onto a page of the SM.

The first step of the using of a file is its DECLARATION and OPENING by the
OPEN instruction. The last step is its CLOSING and if the file is become useless
its DESTRUCTION in both cases by the CLOSE instruction. Between these extreme
steps, READ and WRITE instructions are used.

Two exclusive ACCESS MODES to a file are available in Fortran :

- the SEQUENTIAL ACCESS MODE where all the records are written one after the
other and read one after the other in the same order and with the impossibility to
interlace read and write instructions. A record can not be written or read if its preceding
record has not been (think to the using of a magnetic rubbon).

- the DIRECT ACCESS MODE (or RANDOM access), where any page has the same
size, but readable or writeable in any order. For this, the programmer manages the
PAGE NUMBER on the file with a keyword REC=nb.

Values of variables or arrays are stored on MM or SM on a BINARY or CHARACTER
form. The BINARY form is this used in MM and on registers, without modification but
not readable by a person. The file is a BINARY file.
The CHARACTER form supposes the transformation of the binary of the variables
into characters according to a FORMAT which indicates how to do. The file is a
FORMATTED file.

Remarks :
- For the system, the SCREEN is a SEQUENTIAL FORMATTED file, usable for
WRITING only ( instruction WRITE(*,*) or PRINT *, ).
- For the system, the KEYBOARD is a SEQUENTIAL FORMATTED file for READ-
ING only ( instruction READ(*,*) or READ *, ).
- In both cases, the first character * of WRITE() and READ() designates the UNIT
NUMBER and the second character * designates here the FREE FORMAT or DE-
FAULT FORMAT.
- These two special files are opened and closed directly by the system (which practices
also the re-directions !...)

- It is useful to write the file as “self-readable”, that is to put in the first record all
values which are necessary to read all the other records. That may be the number of
words of each record, the number of records, ...
- A file must constitute a LOGICAL ENTITY as a mesh, a global matrix, a set of
vectors, a drawing, a set of results, ...

The transfer SM-MM is done with 2 steps :
For instance, the variable values are written into a special zone named BUFFER. When
the buffer is full, the central unit delegates to the disk controller the transfer onto a
page of the disk.
This method explains that a not wanted stop of the execution may be prevent a transfer
on the disk, even if the instruction write has been executed !
The instruction READ and WRITE use a BUFFER.
If the operation of Input (instruction read) or Output (instruction write), said I/O, is
FORMATTED, the central unit transforms the binary code into characters before the
transfer.

To explain how to translate the binary code into characters, the programmer may use
SPECIFICATIONS of FORMAT and add texts, “blanks”, ...

The LIST of VARIABLES of the instruction WRITE specifies the storage order of the
values onto the file.

The LIST of VARIABLES of the instruction READ indicates how to separate the
variables of the SM and put the values into the variables in MM.

Of course, the declaration type of variables takes place into this translation, a logical
variable must receive a logical value, a double precision variable must receive a double
precision real value,..., but defines also the bytes size necessary to the storage. 4 bytes
for a real simple precision, 8 bytes for a real double precision, ...

The LIST of VARIABLES is constituted of CONSTANTS, VARIABLES, NAMES of
ARRAYS, separated by a comma, and eventually, framed into 2 brackets, finished by
an integer variable followed by = , the initial value, the final value and the step of an
implicit loop.

More precisely, a small GRAMMAR generates the possibilities to create of a LIST of
VARIABLES :

X :=Y defines the value of X from the value of Y
< > represents a definition seen after or a language primitive
| means OR
{ } signifies as many times as wanted, eventually no times

then

<LIST of VARIABLES> := <Constant> | <Variable> | <Array>

| LIST of VARIABLES { , LIST of VARIABLES }

| ( LIST of VARIABLES, index=IntegerMin, IntegerMax, IntegerStep )

Example :
| | REAL X, A(5), B(2)
| | READ *,X,(A(I),I=1,5,2),B
| | PRINT *,X,(’A(’,I,’)=’,A(I),I=1,5,2),B

requires the reading of X, A(1), A(3), A(5) and of B(1), B(2)

The typing on the keyboard of
0. 10 30.0 -5E1 0.1E+3 2.0E2 3.14159
produces the writing of
0.0 A(1)=10.0 A(3)=30.0 A(5)=-50.0 100.0 200.0

Attention :
- Here, a BLANK CHARACTER is a SEPARATOR of values (This is not the case in
fortran instruction) ;
- the typing line is abandoned after the reading of 2.0E2 and 3.14159 will never be read,
fault of no variable in the list to receive this typed value !

The USING of SEQUENTIAL ACCESS FILES :

- To declare and open a sequential file :
| | OPEN( UNIT=nf, FILE=’FileName’,
|%| STATUS=’NEW’ or ’OLD’ or ’SCRATCH’ or ’UNKNOWN’,
|%| ACCESS=’SEQUENTIAL’ or ’APPEND’,
|%| FORM=’FORMATTED’ or ’UNFORMATTED’,
|%| ERR=Label, IOSTAT=IntegerVariable )

nf is either, a not signed integer, or, an integer variable which takes a positive integer
value. This value must always been present. nf is the UNIT FILE NUMBER.
FileName is the Unix file Name.
In fact, the instruction OPEN transforms the Unix file Name into an integer which is
more practical to handle.

The ”STATUS” indicates the initial file status.

STATUS=’NEW’ specifies a new file to be created.
If the file exists, IOSTAT receives a positive integer value to mean an error and the
execution continues at the instruction with the ERR label if it is present.
Without IOSTAT and ERR, the execution stops.

STATUS=’OLD’ specifies an existing file.
If the file does not exist, IOSTAT receives a positive integer value to mean an error and
the execution continues at the instruction with the ERR label if it is present.
Without IOSTAT and ERR, the execution stops.

STATUS=’SCRATCH’ specifies a new file to be created, used and deleted after
treatment.

STATUS=’UNKNOWN’ accepts a new or existing file.

ACCESS=’SEQUENTIAL’ specifies the mode access mode (This is the default value).

ACCESS=’APPEND’ defines the sequential access mode and locates the writing pointer
at the end of the existing file.

FORM=’FORMATTED’ indicates the value translation from binary to characters or
the inverse. This is the default value.

FORM=’UNFORMATTED’ all transfers are on binary form without translation into
characters.

After the execution of OPEN, the POINTER of READING/WRITING is positioned
at beginning of the file.

- To read a list of variables on the unit file nf :
| | READ( UNIT=nf or *, FMT=FormatLabel or *,
|%| IOSTAT=IntegerVariable, ERR=Label,
|%| END=EndLabel ) LIST of VARIABLES

The first character * represents the keyboard.
The second character * represents a Default Format.

nf is the unit file number defined by the OPEN instruction.
FMT=FormatLabel or * is given only if the file is ’FORMATTED’.
FormatLabel is a reference to a FORMAT instruction with this label.
After the reading of all data of the file, the following executed instruction is those with
the EndLabel. Thus, it is possible to read a file without the knowledge of its record
number.
Remark : READ(*,FormatLabel) List is equivalent to READ FormatLabel, List
The POINTER of READING is after positioned at the beginning of the following record
to read.

- To write a list of variables on the unit file nf :
| | WRITE( UNIT=nf or *, FMT=FormatLabel or *,
|%| IOSTAT=IntegerVariable, ERR=Label ) LIST of VARIABLES

The first character * represents the screen.
The second character * represents a Default Format.
nf is the unit file number defined by the OPEN instruction.
FormatLabel is a reference to a FORMAT instruction with this label.
Remark : WRITE(*,FormatLabel) List is equivalent to PRINT FormatLabel, List
The POINTER of WRITING is after positioned at the beginning of the following record
to write.

Attention :
Any writing on a sequential file loses all the records which follow.
Consequently, to modify a file, it is necessary to
- Open the file1 to modify ;
- Create and open a new file file2 ;
- Loop on the records of file1
– reading of the file1 record
– modification in MM of the values of one or more values of this record ;
– writing of this record on file2 ;
- End of Loop
- Close the file2
- Delete the file1

| | REWIND ( UNIT=nf, IOSTAT=IntegerVariable, ERR=Label )

locates the READING/WRITING POINTER at the beginning of the first record of the
file nf.

If an error is met, the IntegerVariable of IOSTAT contains the not zero error number
and permits its control.
If an error is met and ERR present, then the next executed instruction is this with the
Label.

The POINTER of READING/WRITING may be move back of ONE record by the
instruction

| | BACKSPACE ( UNIT=nf, IOSTAT=IntegerVariable, ERR=Label )

If an error is met, the IntegerVariable of IOSTAT contains the not zero error number
and permits its control.
If an error is met and ERR present, then the next executed instruction is the one with
the Label.

- To close of a file :
| | CLOSE( UNIT=nf, STATUS=’KEEP’ or ’DELETE’,
|%| IOSTAT=IntegerVariable, ERR=Label )

closes the file of unit file number nf and keep it if STATUS=’KEEP’ or delete it if
STATUS=’DELETE’.
If an error is met, the IntegerVariable of IOSTAT contains the not zero error number
and permits its control.
If an error is met and ERR present, then the next executed instruction is the one with
the label.

- To recover file characteristics from the file unit nf or ’FileName’ :
| | INQUIRE( FILE=’FileName’ or UNIT=nf, ERR=Label,
|%| EXIST=LogicalVar, OPENED=LogicalVar,
|%| NAMED=LogicalVar, NAME=Varcharacters,
|%| SEQUENTIAL=CharactersVar, DIRECT=CharactersVar,
|%| FORMATTED=CharactersVar, UNFORMATTED=CharactersVar,
|%| NUMBER=nf, ACCESS=CharactersVar, FORM=CharactersVar )

EXIST=LogicalVar returns .TRUE. or .FALSE. if the file exists or not
OPENED=LogicalVar returns .TRUE. or .FALSE. if the file is opened or not
NAMED=LogicalVar returns .TRUE. or .FALSE. if the file est named or not
NAME=CharactersVar returns the FileName if the file exists
SEQUENTIAL=CharactersVar returns ’YES’ or ’NO’ as the file is or not.
DIRECT=CharactersVar returns ’YES’ or ’NO’ as the file is or not.
FORMATTED=CharactersVar returns ’YES’ or ’NO’ or ’UNKNOWN’ as the file is or
not.
UNFORMATTED=CharactersVar returns ’YES’ or ’NO’ or ’UNKNOWN’ as the file
is or not.
NUMBER=nf returns the unit file number
ACCESS=CharactersVar returns ’SEQUENTIAL’ as the file is or not.
FORM=CharactersVar returns ’FORMATTED’ or ’UNFORMATTED’ as the file is or
not.

The responses are redondent, but it is not necessary to use all the terms.

The USING of DIRECT ACCESS FILES :

Here, main modifications for a direct access file :
- all pages have the same number of bytes ;
- the access is direct thanks to the record of number indicated by the keyword REC !
Remark : Generally, a direct access file is not formatted to evitate the binary-characters
transformations and slow down the computations.

- To declare and open a direct file :

| | OPEN( UNIT=nf, FILE=’FileName’,
|%| STATUS=’NEW’ or ’OLD’ or ’SCRATCH’ or ’UNKNOWN’,
|%| ACCESS=’DIRECT’, FORM=’FORMATTED’ or ’UNFORMATTED’,
|%| RECL=OnePageBytesNumber,
|%| ERR=Label, IOSTAT=IntegerVariable )

nf is either, a not signed integer, or, an integer variable which takes a positive integer
value. This value must always been present. nf is the UNIT FILE NUMBER.
FileName is the Unix file Name.
In fact, the instruction OPEN identifies the Unix file Name into an integer which is
more practical to handle.

The ”STATUS” indicates the initial file status.

STATUS=’NEW’ specifies a new file to be created.
If the file exists, IOSTAT receives a positive integer value to mean an error and the
execution continues at the instruction with the ERR label if it is present.

Without IOSTAT and ERR, the execution stops.

STATUS=’OLD’ specifies an existing file.
If the file does not exist, IOSTAT receives a positive integer value to mean an error and
the execution continues at the instruction with the ERR label if it is present.
Without IOSTAT and ERR, the execution stops.

OnePageBytesNumber is the constant bytes number of each page of the direct file. This
number may be constant or an IntegerVariable initialized with a positive value.

- To read a Page of a direct file :
| | READ( UNIT=nf, FMT=FormatLabel or *,
|%| REC=PageNumber,
|%| IOSTAT=IntegerVariable, ERR=Label ) VARIABLES of the PAGE

REC=PageNumber (integer value greater or equal 1) of which the read values are
distributed in the VARIABLES of the PAGE
FMT=FormatLabel or * is necessar only if the file is ’FORMATTED’.
FormatLabel is a reference to a FORMAT instruction with this label seen later.

- To write a Page of a direct file :
| | WRITE( UNIT=nf or *, FMT=FormatLabel or *,
|%| REC=PageNumber,
|%| IOSTAT=IntegerVariable, ERR=Label ) VARIABLES of the PAGE

- To close a file :
| | CLOSE( UNIT=nf, STATUS=’KEEP’ or ’DELETE’,
|%| IOSTAT=IntegerVariable, ERR=Label )

closes the file of unit file number nf and keep it if STATUS=’KEEP’ but delete it if
STATUS=’DELETE’.
If an error is met, the IntegerVariable of IOSTAT contains the not zero error number
and permits its control.
If an error is met and ERR present, then the next executed instruction is this with the
Label.

- To recover the file characteristics from the unit nf or ’FileName’ :
| | INQUIRE( FILE=’FileName’ or UNIT=nf, ERR=Label,
|%| EXIST=LogicalVariable,
|%| OPENED=LogicalVar, NAMED=LogicalVar, NAME=CharactersVar,
|%| SEQUENTIAL=CharactersVar, DIRECT=CharactersVar,
|%| FORMATTED=CharactersVar, UNFORMATTED=CharactersVar,
|%| NUMBER=nf, ACCESS=CharactersVar, FORM=CharactersVar,
|%| RECL=IntegerVar, NEXTREC=IntegerVar )

The parameters are already described, excepted two new parameters
RECL=IntegerVar returns the bytes number of a page of this file.
NEXTREC=IntegerVar returns the number+1 of the last page read or written before
the execution of the inquire instruction. (The number of the last read or written page
has been incremented of 1)

The FORMAT INSTRUCTION :
The format instruction indicates how the values must be
- read from string of characters, code in a binary form and put in MM variables ;

- write from the binary value of variables into formatted string of characters, blend with
string of characters to form a text pleasant to read and put on the file, which may be a
screen.

Its GENERAL FORM is :

Label || FORMAT( list of specifications )

< list of specifications > := specif1 | specif2 |

| list of specifications {, list of specifications }

| list of specifications { / list of specifications }

| NotSignedIntegerConstant ( list of specifications )

After, m n p are Not Signed Integer Constant as 7 or 25 or ...

< specif1 > := mIn writing or reading of m integers each on n characters

| mFn.p writing or reading of m reals each on n characters with p digits
after the . decimal

| mEn.p writing or reading of m reals each on n characters with p digits
after the . decimal but under a normalized form
sign Y.XXXXXXX E sign Exposant of 10 with Y not zero
it is necessary to choose n≥p+7 to obtain in any case a correct printed value.
p>7 does not give more precision than p=7.

| mDn.p writing or reading of m reals double précision each on n characters
with p digits after the . decimal but under a normalized form
sign Y.XXXXXXX D sign Exposant of 10 with Y not zero
it is necessary to choose n≥p+7 to obtain in any case a correct printed value.
p>17 does not give more precision than p=17.

| mGn.p writing or reading of m reals under the most appropriated form
mFn.p or mEn.p or mDn.p

| mLn writing or reading of m Logical each on n characters
(For input, the first character t or f met gives the value .TRUE. or .FALSE.

| mAn writing or reading of m strings of characters, each on n characters.

< specif2 > := ’STRING’

| mX produces m ”blanks” in the text

| Tn indicates that the next character will be at the n-th column

| TLn displaces the next I/O of n characters towards the left

| TRn displaces the next I/O of n characters towards the right

| / to pass at the next line

| $ to evitate in end of format to pass at the next line.

Example :
100 | | FORMAT( 10I8 )
101 | | FORMAT( ’The MATRIX A =’ / 5(’A(’,I1,’,’,I1,’)=’,E15.6) )
102 | | FORMAT( ’RESULTS :’ / T10, ’Exact VALUE’, T30, ’Computed VALUE’/

| | (T10, G15.5, T30, G15.5 ) )

The LINK between the LIST of VARIABLES and the FORMAT instruction

The variable of the list, written in extension, must fit to the specifications of type 1,
themselves written in extension :
The variable1 is associated to the first specif1,
the variable2 is associated to the second specif1, ...

3 CASES are POSSIBLE :

1. The number of variables is equal to the number of specif1 :
=> correspondence variable-specif1 and pass to to the next line

2. There are less variables than specif1 :
=> abandonment of the format at the first specif1 which has no variable to fit and pass
to the next line

3. There are more variables than specif1 :
=> at end of format, pass to the next line and generate again the format
. at its beginning if the format has no brackets
. at the beginning of the bracket couple, which is not itself between brackets and the most
right located after extension (repetition factor included) and this, up to all variables hit
a specif1.

Example :
- For a reading, use the FREE FORMAT * , the simplest way !

PROGRAM LECECR
C Reading and Writing wit the FORMAT instruction

REAL A(1 :5)
C
C ASK the COMPONENT NUMBER of the VECTOR A

PRINT *,’Vector Component NUMBER = ?’
READ *,N
DO I=1,N

A(I) = I*I
ENDDO

C
PRINT 10010,(I,A(I),I=1,N)

10010 FORMAT(’The Result VECTOR’/
% 2(’ A(’,I1,’)=’,G15.6) )
STOP
END

3 executions of this file with the reading of N=1, then 2, then 5, gives the text
Vector Component NUMBER =

1
The Result VECTOR
A(1)= 1.00000 A(

Vector Component NUMBER =
2
The Result VECTOR
A(1)= 1.00000 A(2)= 4.00000

Vector Component NUMBER =
5
The Result VECTOR
A(1)= 1.00000 A(2)= 4.00000
A(3)= 9.00000 A(4)= 16.0000
A(5)= 25.0000 A(

The FORMAT has been generated again twice at the beginning of its second line.

Exercise 1 :
Print matrices A(1 :3,1 :3), B(4,4), C(5,5) with a same FORMAT instruction.

A solution : (Read it after you have not succeed to do the exercise !)

PROGRAM EX1
REAL A(3,3), B(4,4), C(5,5)
CALL INITM( 3, A )
CALL INITM( 4, B )
CALL INITM( 5, C )

C
CALL IMPMAT( ’A’, A, 3 )
CALL IMPMAT( ’B’, B, 4 )
CALL IMPMAT( ’C’, C, 5 )
STOP
END

C
SUBROUTINE INITM( N, A )

C INITIALIZATION of the MATRIX A
REAL A(N,N)
DO 10 I=1,N

DO 5 J=1,N
A(I,J) = I*J

5 CONTINUE
10 CONTINUE

RETURN
END

C
SUBROUTINE IMPMAT( K, X, N )

C Printing of the MATRIX K(N,N) DEFINED in X
REAL X(N,N)
CHARACTER*(*) K

C
WRITE(*,*)
WRITE(*,*) ’The MATRIX ’,K, ’ :’
DO 10 I=1,N

WRITE(*,10010) (K,I,J,X(I,J),J=1,N)
10 CONTINUE

10010 FORMAT( 5(A1,’(’,I1,’,’,I1,’)=’,G15.5) )
C

RETURN
END

Numerical Results :

The MATRIX A :
A(1,1)= 1.0000 A(1,2)= 2.0000 A(1,3)= 3.0000
A(2,1)= 2.0000 A(2,2)= 4.0000 A(2,3)= 6.0000
A(3,1)= 3.0000 A(3,2)= 6.0000 A(3,3)= 9.0000

The MATRIX B :
B(1,1)= 1.0000 B(1,2)= 2.0000 B(1,3)= 3.0000 B(1,4)= 4.0000
B(2,1)= 2.0000 B(2,2)= 4.0000 B(2,3)= 6.0000 B(2,4)= 8.0000
B(3,1)= 3.0000 B(3,2)= 6.0000 B(3,3)= 9.0000 B(3,4)= 12.000
B(4,1)= 4.0000 B(4,2)= 8.0000 B(4,3)= 12.000 B(4,4)= 16.000

The MATRIX C :
C(1,1)= 1.0000 C(1,2)= 2.0000 C(1,3)= 3.0000 C(1,4)= 4.0000 C(1,5)= 5.0000
C(2,1)= 2.0000 C(2,2)= 4.0000 C(2,3)= 6.0000 C(2,4)= 8.0000 C(2,5)= 10.000
C(3,1)= 3.0000 C(3,2)= 6.0000 C(3,3)= 9.0000 C(3,4)= 12.000 C(3,5)= 15.000
C(4,1)= 4.0000 C(4,2)= 8.0000 C(4,3)= 12.000 C(4,4)= 16.000 C(4,5)= 20.000
C(5,1)= 5.0000 C(5,2)= 10.000 C(5,3)= 15.000 C(5,4)= 20.000 C(5,5)= 25.000

9



Exercise 2 :
Write the lower triangular part of a matrix A(1 :N,1 :N).

A solution :

PROGRAM EX2
REAL A(3,3), B(4,4), C(5,5)
CALL INITM( 3, A )
CALL INITM( 4, B )
CALL INITM( 5, C )

C
CALL IMPMTI( ’A’, A, 3 )
CALL IMPMTI( ’B’, B, 4 )
CALL IMPMTI( ’C’, C, 5 )
STOP
END

C
SUBROUTINE INITM( N, A )

C INITIALIZATION of the MATRIX A
REAL A(N,N)
DO 10 I=1,N

DO 5 J=1,I
A(I,J) = I*J

5 CONTINUE
10 CONTINUE

RETURN
END

SUBROUTINE IMPMTI( K, X, N )
C WRITING of the LOWER TRIANGULAR PART of a MATRIX
C The MATRIX K(N,N) is DEFINED in X

REAL X(N,N)
CHARACTER*(*) K

C
WRITE(*,*)
WRITE(*,*) ’The MATRIX ’,K
DO 10 I=1,N

WRITE(*,10010) (K,I,J,X(I,J),J=1,I)
10 CONTINUE

10010 FORMAT( 5(A1,’(’,I1,’,’,I1,’)=’,G15.5) )
RETURN
END

Numerical Results :

The MATRIX A
A(1,1)= 1.0000
A(2,1)= 2.0000 A(2,2)= 4.0000
A(3,1)= 3.0000 A(3,2)= 6.0000 A(3,3)= 9.0000

The MATRIX B
B(1,1)= 1.0000
B(2,1)= 2.0000 B(2,2)= 4.0000
B(3,1)= 3.0000 B(3,2)= 6.0000 B(3,3)= 9.0000
B(4,1)= 4.0000 B(4,2)= 8.0000 B(4,3)= 12.000 B(4,4)= 16.000

The MATRIX C
C(1,1)= 1.0000
C(2,1)= 2.0000 C(2,2)= 4.0000
C(3,1)= 3.0000 C(3,2)= 6.0000 C(3,3)= 9.0000
C(4,1)= 4.0000 C(4,2)= 8.0000 C(4,3)= 12.000 C(4,4)= 16.000
C(5,1)= 5.0000 C(5,2)= 10.000 C(5,3)= 15.000 C(5,4)= 20.000 C(5,5)= 25.000

ADVANCED USING of the FORMAT instruction :
The unit file number nf is replaced here by a STRING of CHARACTERS and the file
is said to be an “internal file”.
No OPEN or CLOSE instruction to do.

| | READ( UNIT=CharactersVariables,
|%| FMT=Label | ’(The Specifs )’ | Characters Variable or Expression,
|%| ERR=Label, END=Label ) LIST of VARIABLES

| | WRITE( UNIT=CharactersVariables,
|%| FMT=Label | ’(The specifs )’ | Variable or Expression characters,
|%| ERR=Label ) LIST of VARIABLES

In fact, in place of the definition of the string of characters of the FORMAT instruction,
this string is directly written after FMT= between ’ ’.

Exercise 3 :
Write the upper triangular part of a matrix A(N,N) with N ≤ 5)

A solution :

PROGRAM EX3
REAL A(2,2), B(3,3), C(4,4)
CALL INITM( 2, A )
CALL INITM( 3, B )
CALL INITM( 4, C )

C
CALL IMPMTS( ’A’, A, 2 )
CALL IMPMTS( ’B’, B, 3 )
CALL IMPMTS( ’C’, C, 4 )
STOP
END

C
SUBROUTINE INITM( N, A )

C INITIALIZATION of the MATRIX A
REAL A(N,N)
DO 10 I=1,N

DO 5 J=I,N
A(I,J) = I*J

5 CONTINUE
10 CONTINUE

RETURN
END

SUBROUTINE IMPMTS( K, X, N )
C WRITING of the UPPER TRIANGULAR PART of a MATRIX
C The MATRIX K(N,N) is DEFINED in X

REAL X(N,N)
CHARACTER*(*) K
CHARACTER*42 F

C
WRITE(*,*)
WRITE(*,*) ’The MATRIX ’,K,’ :’
DO I=1,N

C START POSITION of the WRITING
WRITE(F(1 :2),’(I2)’) 22*I-21

C MATRIX COEFFICIENTS NUMBER to WRITE
WRITE(F(3 :3),’(I1)’) N-I+1

C F CONTAINS the STRING of CHARACTERS which CODES the NUMBER
F=’(T’//F(1 :2)//’,’//F(3 :3)//

% ’(1X,A1,’’(’’,I1,’’,’’,I1,’’)=’’,E13.6,1X))’
PRINT *,F(1 :42)
WRITE(*,FMT=F(1 :42)) (K,I,J,X(I,J),J=I,N)

ENDDO
C

RETURN
END

Numerical Results :

The MATRIX A :
(T 1,2(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))
A(1,1)= 0.100000E+01 A(1,2)= 0.200000E+01
(T23,1(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

A(2,2)= 0.400000E+01

The MATRIX B :
(T 1,3(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))
B(1,1)= 0.100000E+01 B(1,2)= 0.200000E+01 B(1,3)= 0.300000E+01
(T23,2(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

B(2,2)= 0.400000E+01 B(2,3)= 0.600000E+01
(T45,1(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

B(3,3)= 0.900000E+01

The MATRIX C :
(T 1,4(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))
C(1,1)= 0.100000E+01 C(1,2)= 0.200000E+01 C(1,3)= 0.300000E+01 C(1,4)= 0.400000E+01
(T23,3(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

C(2,2)= 0.400000E+01 C(2,3)= 0.600000E+01 C(2,4)= 0.800000E+01
(T45,2(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

C(3,3)= 0.900000E+01 C(3,4)= 0.120000E+02
(T67,1(1X,A1,’(’,I1,’,’,I1,’)=’,E13.6,1X))

C(4,4)= 0.160000E+02

Exercise 4 :
Draw with characters on the screen defined by 25 lines of 80 characters, the function
cosine(x) for x∈ [0, π].

A solution :

PROGRAM EX4
CALL TRACOS
STOP
END

C
SUBROUTINE TRACOS
PARAMETER (NL=25, NC=80)
CHARACTER*(NC) SCREEN(NL)

C
C Initialization and Y-axis

DO 1 J=1,NL
SCREEN(J) = ’ ’
SCREEN(J)(1 :1) = ’|’

1 CONTINUE
SCREEN(1)(1 :2) = ’+1’
SCREEN(NL)(1 :2) = ’-1’

C
C The X-axis

NL2 = NL / 2 + 1
DO 2 I=1,NC

SCREEN(NL2)(I :I) = ’-’
2 CONTINUE

SCREEN(NL2)(1 :1) = ’O’
SCREEN(NL2)(NC-1 :NC) = ’>X’

C
C Filling of the CURVE POINTS

PAS = ATAN(1.0) * 4.0 / (NC-1)
X = 0.0
DO 3 I=1,NC

SCREEN(NL2-(NL/2)*COS(X))(I :I) = ’*’
X = X + PAS

3 CONTINUE
C
C Write the SCREEN

print *,’COS(X)’
DO 4 I=1,NL

PRINT 10004,SCREEN(I)
4 CONTINUE

10004 FORMAT(A)
C

RETURN
END

Numerical Results :

The precision may be improved !
Remark the suppresion of +1 by the 2 first values *.

COS(X)
***********
| ****
| ****
| ***
| **
| ***
| **
| **
| ***
| **
| **
| **
O---------------------------------------**------------------------------------>X
| **
| **
| ***
| **
| **
| ***
| **
| ***
| ****
| ****
| ***********
-1

13


